

Belant, J. L., Bled, F., Mwampeta, S. B., Mkasanga, I. J., Wilton, C. M. and Fyumagwa, R. 2017. Temporal and spatial variation of broadcasted vocalizations does not reduce lion *Panthera leo* habituation. – Wildlife Biology 2017: wlb.00287

Appendix 1

Winbugs code used to model effects of temporal and spatial variation of broadcasted vocalizations on lion *Panthera leo* habituation

```

model {

##### CALLS #####

## Priors

# Ecological process
intercept.abun.c ~ dnorm(0, 0.01)

# Random effects
for (i in 1:n.site.c) { # Loop over sites
  u.abun.c[i] ~ dnorm(0, tau.N.c)

# Observation process
for (t in 1:n.week) { # Loop over surveys
  p.detect[i,t] ~ dunif(0, 1)
}
}

tau.N.c <- pow(sigma.N.c, -2)
sigma.N.c ~ dunif(0, 5)

## State process
for (i in 1:n.site.c) { # Loop over sites
  N.c[i] ~ dpois(lambda.c[i])
  log(lambda.c[i]) <- max(-10, min(loglam.c[i], 10))
  loglam.c[i] <- (intercept.abun.c + u.abun.c[i])/10
}

## Observation process
for (k in 1:n.record) { # Loop over surveys
  # Abundance
  y.c[site[k], week[k]] ~ dbin(p.detect[site[k], week[k]], N.c[site[k]])
}

```

```

for (l in 1:n.record.na) {
  y.c[site.na,week.na] ~ dbin(.5,10)
  Y.new.c[site.na,week.na] ~ dbin(.5,10)
}

## Derived quantities
totalN.c <- sum(N.c[]) # Population size over all R sites

for(t in 1:n.week){
  p.detect.est[t,1]<-mean(p.detect[1:10,t])
  p.detect.est[t,2]<-mean(p.detect[11:20,t])
  p.detect.est[t,3]<-mean(p.detect[21:30,t])
}

# GOF
for (k in 1:n.record) { # Loop over surveys

  # Compute fit statistics for observed data
  eval.c[site[k],week[k]]<-p.detect[site[k],week[k]]*N.c[site[k]]
  E.c[k]<- pow((y.c[site[k],week[k]]-eval.c[site[k],week[k]]),2)/(eval.c[site[k],week[k]]+0.5)

  # Generate replicate data and compute fit stats for them
  Y.new.c[site[k],week[k]] ~ dbin(p.detect[site[k],week[k]],N.c[site[k]])
  E.new.c[k]<- pow((Y.new.c[site[k],week[k]]-eval.c[site[k],week[k]]),2)/(eval.c[site[k],week[k]]+0.5)

}

fit.c<-sum(E.c[])
fit.new.c<-sum(E.new.c[])

}

```